Usean yhtälön ryhmän ratkaiseminen yhteen yhtälöön palauttamalla

Olkoon tarkasteltavana kolmen differentiaaliyhtälön normaalimuotoinen ryhmä, jossa tuntemattomina funktioina ovat $x(t)$, $y(t)$ ja $z(t)$:

```
dy1 = x'[t] ==  x[t] + y[t] - z[t] + 1;

dy2 = y'[t] == - x[t] + 5 y[t] + z[t] + t;

dy3 = z'[t] == - 2 x[t] + 2 y[t] + 4 z[t] + t^2;

{dy1, dy2, dy3} // TableForm
```

$$x'[t] == 1 + x[t] + y[t] - z[t]$$
$$y'[t] == t - x[t] + 5 y[t] + z[t]$$
$$z'[t] == t^2 - 2 x[t] + 2 y[t] + 4 z[t]$$

```
funktiot = {x[t], y[t], z[t]};
```

Ryhmä pyritään ratkaisemaan eliminoimalla ensin yhtälöistä funktiot x ja y, jolloin saadaan yksinomaan funktiota z koskeva differentiaaliyhtälö. Tätä varten kaksi ensimmäistä yhtälöä derivoidaan kerran ja viimeinen yhtälö kaksi kertaa, jolloin saadaan kaikkiaan seitsemän yhtälöä:

```
dy4 = D[dy1, t];

dy5 = D[dy2, t];

dy6 = D[dy3, t];

dy7 = D[dy3, {t, 2}];

{dy1, dy2, dy3, dy4, dy5, dy6, dy7} // TableForm
```

$$x'[t] == 1 + x[t] + y[t] - z[t]$$
$$y'[t] == t - x[t] + 5 y[t] + z[t]$$
$$z'[t] == t^2 - 2 x[t] + 2 y[t] + 4 z[t]$$
$$x''[t] == x'[t] + y'[t] - z'[t]$$
$$y''[t] == 1 - x'[t] + 5 y'[t] + z'[t]$$
$$z''[t] == 2 t - 2 x'[t] + 2 y'[t] + 4 z'[t]$$
$$z^{(3)}[t] == 2 - 2 x''[t] + 2 y''[t] + 4 z''[t]$$

Kuudesta ensimmäisestä yhtälöstä voidaan ratkaista funktiot x ja y derivaattoineen ja tämän jälkeen sijoittaa lausekkeet perättäin viimeiseen yhtälöön:

```
elimsij = Solve[{dy1, dy2, dy3, dy4, dy5, dy6}, {x[t], y[t], x'[t], y'[t], x''[t], y''[t]}]
```

$$\{\{x[t] \to -\frac{1}{4}(-2+4t-4t^2-12z[t]+8z'[t]-z''[t]),$$
$$y[t] \to -\frac{1}{4}(-2+4t-2t^2-4z[t]+6z'[t]-z''[t]),$$
$$x'[t] \to -\frac{1}{2}(-4+4t-3t^2-6z[t]+7z'[t]-z''[t]),$$
$$y'[t] \to -\frac{1}{2}(-4+6t-3t^2-6z[t]+11z'[t]-2z''[t]),$$
$$x''[t] \to -\frac{1}{2}(-8+10t-6t^2-12z[t]+20z'[t]-3z''[t]),$$
$$y''[t] \to -\frac{1}{2}(-18+26t-12t^2-24z[t]+46z'[t]-9z''[t])\}\}$$

```
dy = dy7 /. First[elimsij]
```

$$z^{(3)}[t] == 12-16t+6t^2+12z[t]-26z'[t]+10z''[t]$$

Tuloksena on kolmannen kertaluvun yhtälö funktiolle z. Tämä voidaan löytää myös hieman suorempaan käyttämällä Mathematican `Eliminate`-komentoa:

```
Eliminate[{dy1, dy2, dy3, dy4, dy5, dy6, dy7}, {x[t], y[t], x'[t], y'[t], x''[t], y''[t]}]
```

$$z^{(3)}[t] == 12-16t+6t^2+12z[t]-26z'[t]+10z''[t]$$

Yhtälön funktiomuotoiseksi ratkaisuksi saadaan

```
zratk = DSolve[dy, z, t]
```

$$\{\{z \to \text{Function}[\{t\}, \frac{1}{36}(-71-30t-18t^2)+e^{(2-\sqrt{2})t}C[1]+e^{(2+\sqrt{2})t}C[2]+e^{6t}C[3]]\}\}$$

Yhtälöryhmän yleinen ratkaisu muodossa $x(t)$, $y(t)$, $z(t)$ saadaan tätä ja aiempia lausekkeita käyttäen:

```
ylratk = funktiot /. First[elimsij] /. First[zratk] // Simplify;
ylratk // TableForm
```

$$\frac{1}{2}\left(-8-3t-t^2+\left(1+2\sqrt{2}\right)e^{-(-2+\sqrt{2})t}C[1]+\left(1-2\sqrt{2}\right)e^{(2+\sqrt{2})t}C[2]\right)$$
$$-\frac{17}{36}-\frac{t}{3}+\frac{1}{2}\left(-1+\sqrt{2}\right)e^{-(-2+\sqrt{2})t}C[1]-\frac{1}{2}\left(1+\sqrt{2}\right)e^{(2+\sqrt{2})t}C[2]+e^{6t}C[3]$$
$$-\frac{71}{36}-\frac{5t}{6}-\frac{t^2}{2}+e^{-(-2+\sqrt{2})t}C[1]+e^{(2+\sqrt{2})t}C[2]+e^{6t}C[3]$$

Tiettyä alkuehtoa vastaavat vakioiden arvot saadaan tämän jälkeen tavalliseen tapaan ratkaisemalla algebrallinen yhtälöryhmä. Olkoon alkuehtona $x(0) = 1$, $y(0) = -2$, $z(0) = 3$. Tällöin

```
vakiot = Solve[(ylratk /. t -> 0) == {1, -2, 3}, {C[1], C[2], C[3]}] // Simplify
```

$$\{\{C[3] \to -\frac{43}{252}, C[1] \to \frac{20+23\sqrt{2}}{4+8\sqrt{2}}, C[2] \to \frac{1}{28}\left(72-17\sqrt{2}\right)\}\}$$

Vastaava yksittäisratkaisu saadaan sijoittamalla vakioiden arvot yleiseen ratkaisuun:

```
yksittratk = ylratk /. First[vakiot] // Simplify;
yksittratk // TableForm
```

$$\frac{1}{8} e^{-\left(-2+\sqrt{2}\right) t} \left(20+23 \sqrt{2}+\left(20-23 \sqrt{2}\right) e^{2 \sqrt{2} t}-4 e^{\left(-2+\sqrt{2}\right) t} \left(8+3 t+t^2\right)\right)$$
$$\frac{1}{504} \left(-86 e^{6 t}-\frac{63 \left(-26+3 \sqrt{2}\right) e^{-\left(-2+\sqrt{2}\right) t}}{1+2 \sqrt{2}}-9 \left(38+55 \sqrt{2}\right) e^{\left(2+\sqrt{2}\right) t}-14 (17+12 t)\right)$$
$$-\frac{71}{36}-\frac{43 e^{6 t}}{252}+\frac{\left(20+23 \sqrt{2}\right) e^{-\left(-2+\sqrt{2}\right) t}}{4+8 \sqrt{2}}+\frac{1}{28} \left(72-17 \sqrt{2}\right) e^{\left(2+\sqrt{2}\right) t}-\frac{5 t}{6}-\frac{t^2}{2}$$

Ratkaisujen kuvaajat:

```
Plot[Evaluate[yksittratk], {t, 0, 1}, PlotRange -> {-20, 20}]
```




```
- Graphics -
```

Edellä oleva esitys kuvaa, miten eliminointiprosessi ja yhtälöryhmän ratkaiseminen tapahtuu. Jos tavoitteena on ainoastaan saada tietyn alkuarvoprobleeman ratkaisu, päästään paljon vähemmällä kohdistamalla Mathematican `DSolve`-funktio suoraan alkuperäiseen yhtälöryhmään ja alkuehtoon:

```
suoraratk = funktiot /. Flatten[
    DSolve[{dy1, dy2, dy3, x[0] == 1, y[0] == -2, z[0] == 3}, funktiot, t]] // Simplify;
suoraratk // TableForm
```

$$\frac{1}{8} e^{-\sqrt{2} t} \left(\left(20+23 \sqrt{2}\right) e^{2 t}+\left(20-23 \sqrt{2}\right) e^{2 \left(1+\sqrt{2}\right) t}-4 e^{\sqrt{2} t} \left(8+3 t+t^2\right)\right)$$
$$\frac{1}{504} e^{-\sqrt{2} t} \left(9 \left(-38+55 \sqrt{2}\right) e^{2 t}-9 \left(38+55 \sqrt{2}\right) e^{2 \left(1+\sqrt{2}\right) t}-86 e^{\left(6+\sqrt{2}\right) t}-14 e^{\sqrt{2} t} (17+12 t)\right)$$
$$\frac{1}{252} e^{-\sqrt{2} t} \left(9 \left(72+17 \sqrt{2}\right) e^{2 t}-9 \left(-72+17 \sqrt{2}\right) e^{2 \left(1+\sqrt{2}\right) t}-43 e^{\left(6+\sqrt{2}\right) t}-7 e^{\sqrt{2} t} \left(71+30 t+18 t^2\right)\right)$$

Eri tavoilla saadut ratkaisut ovat todellakin samat:

```
yksittratk == suoraratk // Simplify
```

```
True
```

Linkit

differentiaaliyhtälöryhmä
ryhmän palauttaminen yhteen yhtälöön
differentiaaliyhtälön ratkaiseminen symbolisella ohjelmalla (symalg.nb)

SKK 30.04.2001